Question: Being new to Elixir I'm having some problems understanding pattern matching.

If I have an Elixir data structure like this:

```
list_with_maps = [%{"id" => 1, "name" => "a"}, %{"id" => 2, "name" => "b"}]

```

What would be the best way to get values of all id fields from it?

Answer: 
For sake of completeness for the answers for this question you could also do something like this:

```
defmodule Test do
  def get_all_ids([head | tail ]) do
    IO.puts head["id"]
    get_all_ids(tail)
  end

  def get_all_ids([]) do
    IO.puts "end"
  end
end

```

Which would be used like so:

```
iex(7)> Test.get_all_ids(list_with_maps)
1
2
end
:ok

```

Although I think @Gazler's answer is the better answer in this case.

Oh and since you specifically mentioned pattern matching, this would also work:

```
defmodule Test do
  def get_all_ids([%{"id" => id} = m  | tail ]) do
    IO.puts id
    get_all_ids(tail)
  end

  def get_all_ids([]) do
    IO.puts "end"
  end
end

```

The call would be exactly the same; the difference in the second approach is that it's using a pattern match to parse the map in the argument list. 

You might also change the argument list in this line: `def get_all_ids([%{"id" => id} = m | tail ]) do` to this: `def get_all_ids([%{"id" => id} = _m | tail ]) do` just to avoid the warning about `m` being unused.
Another option is to use the `get_in` function and leveraging the `Access` module. The Access module lets you build expressive ways of dealing with nested data structures.

To get a list of values at `"id"` you could run:

```
get_in(list_with_maps, [Access.all, "id"])  

```

Passing `Access.all` as the first element in the list, followed by `"id"` tells `get_in` to return all `list_with_maps`'s `"id"` value.

It may seem like overkill for this example, but if your structure looked like:

```
list_with_maps = [%{"id" => 1, "identities" => [%{"name" => "a"}, %{"name" => "Secret A"}]}, %{"id" => 2, "identities" => [%{"name" => "b"},%{"name" => "Secret B"}]}]

```

We could get all of the names with:

```
get_in(list_with_maps, [Access.all, "identities", Access.all, "name"])
# [["a", "Secret A"], ["b", "Secret B"]]

```